
 KANDÓ KÁLMÁN DIGITÁLIS TECHNIKA
VILLAMOSMÉRNÖKI FŐISKOLAI KAR 8051 mikrokontroller mérések
AUTOMATIKA INTÉZET

2.MÉRÉS 1.oldal

2.MÉRÉS

PBC80C552 mikrokontroller-bázisú mikrogép programozása

1. A mérés célja:
Az XE2 fejlesztői környezet megismerése. A mikrogép hardver felépítésének, monitor programjá-
nak használata. Programfejlesztés, programírás - fordítás - szerkesztés - futtatás folyamat gya-
korlása.

2. A szükséges ismeretek

2.1. A mikrogép hardver kialakítása
A mérési gyakorlatok a 80C552 típusú mikrokontrolleres gyakorló készüléken kell elvégezni. A
készülék részletes leírása az MK_GY.DOC fájlban van.

2.2. Szoftver rendszer és fejlesztői környezet
A mikrokontrolleres mikrogép - PC osztott intelligenciájú rendszer. A programfejlesztéshez az
XE2 menüvezérelt fejlesztői környezetet használjuk amely a C:/XE251 alkönyvtárban van. Indí-
tása az XE251.BAT parancsfájllal történik. A környezetben mind assembly (*.a51), mind pedig C
(*.c) nyelven végezhető a programfejlesztés.

Az indítás után megjelenő képen, az alábbi hat menüpont között lehet választani.

Fájl Editor Assembler/Compiler Tesztelés Súgó Kilépés/Quit
A menüpontok kiválasztása mind az egérrel, mind pedig a cursor vezérlő nyilakkal történhet. (A
fejlesztői környezet használatának részletes leírása a PROG_FEJL.DOC állományban található.)
Itt röviden leírtuk, hogy az egyes menükben milyen feladatok végezhetők.

A File menün belül a

• Betölt almenüben választható ki a fejlesztendő - már létező - forrásnyelvi fájl,

•  *.A51  assembly nyelvű
•  *.C C nyelvű

formája, illetve a

• *.PRJ project fájl.

• *.* tetszőleges kiterjesztésű állomány.

• Megtekint almenüben tekinthetők meg a fordítás során létrehozott

•  Hiba-fájl
•  Lista-fájl
•  Map-fájl
•  Kevert C lista

• Takarít almenüben törölhetők a fejlesztéskor létrejött fájlok.
• Kilépés/Quit almenün keresztül léphetünk ki a fejlesztői környezetből.

Az Edit menün belül választható ki, majd irható, módosítható

•  Forrás a kiválasztott forrásnyelvi fájl.
•  Project a project fájl.

 Az Assembler/Compiller menüben választható ki

•  Fordítás a forrás-fájlok assemblálása, ill. compillálása.(Ekkor készülnek el a
*.OBJ - tárgykódú,  és a *.LST – lista fájlok.

•  Linker szerkesztés, amely során az abszolút-object (futtatható), ill. a
*.MAP  memória-térkép fájlok generálódnak.

•  Hex generálás itt történik a futatható fájl letölthető formájúvá *.HEX
alakítása.

 A Linker almenü kiválasztásakor egy további ablakban lehet kiválasztani az

•  Egyedi alpontban az egy modulos assembly-nyelvű,

•  Egyedi+Startup alpontban az egy modulos C-nyelvű,

•  Project alpontban a *.PRJ parancsfájllal több modulból álló programok
szerkesztését.

 A Tesztelés menüben végezhető el az előzőek során hibátlannak bizonyult program ellenőrzése
a

•  Szimulátor

almenüben szimulálása a Simula5x program segítségével. (Használata az
1. mérési útmutatóban található).

• Terminál

almenüben a valós-idejű ellenőrzés végezhető. A menüpont elindításakor
jön létre - soros vonalon keresztül - a kommunikációs kapcsolat a PC és a
mikrogép között. A kapcsolatot a PC-n futó TVTERM.EXE, és a
mikrogépben futó monitor és tesztelő programok hozzák lét-
re.(Használatát a  Mon_haszn.doc segédanyag írja le részletesen ).

•  Disassembler megnyitásakor újabb ablakból választható ki

•  Nyers forrás a futtatható kódsorozat illetve,
•  C-hivatkozásokkal C nyelvű forrásnál a kevert

visszafordítás ( disassemblálás, vagy unassemblálás).

 A Sugó menü alatt különböző segédanyagok, pl. utasítás-lista stb. olvashatók.

2.3. A mikrogép monitor és tesztelő programja
A mikrogép EPROM - jában van a monitor és tesztelő program. A monitor segíti a fejlesztett
program futtatását, ellenőrzését. Biztosítja a soros kommunikációt a terminálként használt PC-vel.
Parancsai - a terminál funkcióban - a képernyőre kiírathatók a H gomb lenyomásával.

3. Házi feladatok

3.1. Elemezze az alábbi ELSO_PR.A51 nevű programot, amely a mikrokontrolleres gyakorlón ki-
épített nyomógombok, és LED kijelzők használatára mutat példát! Készítse el a program fo-
lyamatábráját!

;***********************************************************
;* BMF KKVFK  Automatika Intézet *
;*   Elektronika Szakcsoport *
;***********************************************************
;*  Főprogram: Első program *
;*  Változat: V1.0 *
;*  Dátum: 2003.07.20 *
;*  Készítette: Zalotay Péter *
;***********************************************************
;*  A modul leírása: *
;*  Az MK_GY mikrokontrolleres gyakorlón lévő *
;* LED-ek és nyomgombok hasznlata futó-fény *
;* vezérlésére *
;***********************************************************

NAME Elso_prog ;a program modul neve

;xxxxxxx DEKLARÁCIÓK xxxxxxx

;**** Konstansok deklarációja *****

IDO1 EQU 20H ;a külső ciklus ismétlési száma
IDO2 EQU 0A0H ;a belső ciklus ismétlési száma

;**** SZEGMENSEK DEKLARÁCIÓJA *****

;++++ Relokálható szegmensek ++++

VALT SEGMENT DATA ;adatszegmens bájtos változóknak

VALTB SEGMENT DATA BITADDRESSABLE
;adatszegmens a bitcímezhető
;területen lévő változóknak

?STACK SEGMENT IDATA ;adatszegmens a ZSÁK-mem.részére
PROG SEGMENT CODE ;program szegmens

;++++ Abszolút címzésű szegmensek ++++

XSEG AT 0C000H ;a memóriába  ágyazott
;perifériák
;külső memória-területen

NGS: DS 1 ;nyomógomb-sor címe 0C000H
LEDS: DS 1 ;LED-sor címe 0C001H

;++++ Bájtos változók szegmense ++++

RSEG VALT
DS 2 ; helyfoglalás 2 bájt részére (csak példaként)

KIM: DS 1 ; helyfoglalás  a kimeneti memória részére
M_VALT: DS 1 ; helyfoglalás a munka változó részére

;++++ Bit-címezhető változók szegmense ++++

RSEG VALTB

DS 2 ; helyfoglalás 2 bájt részére (csak példaként)
BEM: DS 1 ; helyfoglalás a bemeneti memória részére
ATM: DS 1 ; helyfoglalás az átmeneti tároló részére

;  a bájtok egyes bitjeihez szimbólum rendelése

BE_NG BIT BEM.0 ;be-nyomógomb a BEM bájt LSB bit-je
KI_NG BIT BEM.7 ;ki-nyomógomb a BEM bájt MSB bit-je
TART BIT ATM.0 ;tartó-bit az ATM  bájt LSB bit-je

;++++ A zsák (stack) szegmense ++++

RSEG ?STACK

DS 10H

;++++ A program szegmense ++++

RSEG PROG

LJMP INIT ;ugrás az inicializálásra

DS 10H ;  megszakítás-ugrótáblázatnak helyfoglalás

;xxxx INICIALIZÁLÁS xxxx

INIT: MOV SP,#?STACK-1 ;az SP beállítása

;.... memória-nullázó for-ciklus ...
;   ciklus-jellemzők megadása

MOV R0,#VALT ;első bájt címének megadása
MOV R7,#20H ;ciklus-változó a bájtok száma

;   ciklus-törzs
ISM: MOV @R0,#0 ;a címzett bájtba 0 írása

INC R0 ;következő cím

;   tesztelés

DJNZ R7,ISM ;a ciklusváltozó csökkentése, és végellenőrzés
;   kezdő-értékek beállítása

MOV M_VALT,#80H ;a munka változó MSB-bitje 1

MOV DPTR,#LEDS ;a LED-sor sötétre vezérlése
MOV A,#0FFH
MOVX @DPTR,A

;xxxx FŐPROGRAM xxxx

FOPR:
;.... BEOLVAS ....

MOV DPTR,#NGS ;a nyomógombok aktuális értékének
MOVX A,@DPTR ;beolvasása a BEM memóriába
CPL A
MOV BEM,A

;.... VEZÉRLÉS ....

MOV C,BE_NG ;a TART-bit vezérlése a
ORL C,TART ;           ______
ANL C,/KI_NG ;TART = (BE_NG+TART)*KI_NG
MOV TART,C ;logikai függvény szerint;(öntartás)

;.... DÖNTÉS ....

JNB TART,VEGE ;ha a TART nem 1, akkor áll a világító LED

;.... FORGATÁS ....

MOV A,M_VALT ;TART=1 esetén a munka- változó tartalmának
RL A ; forgatása balra
MOV M_VALT,A

;.... KIIRATÁS ....

MOV DPTR,#LEDS ;az M_VALT tartalmának kijelzése a LED-soron
MOV A,M_VALT
CPL A
MOVX @DPTR,A

;.... KÉSLELTETÉS ....

CALL KESL ;a forgatás ismétlésének késleltetése

VEGE:
JMP FOPR

;xxxx ALPROGRAMOK (szubrutinok) xxxx

;.... Szoftver-késleltető rutin ....
; egymásba ágyazott ciklusokkal, az R7 és R6 ciklusváltozók

KESL:
MOV R7,#IDO1 ;kezdőértékek beállítása

KES1:
MOV R6,#IDO2

KES2:
DJNZ R6,KES2 ;belső ciklus
DJNZ R7,KES1 ;külső ciklus
RET ;visszatérés a főprogramba

END

4. Mérési feladatok:

4.1. Másolja át az C:\ XE251\MUNKA könyvtáron belül létrehozott saját alkönyvtárába  a
PELDAK alkönyvtárban található  ELSO_PR. A51 példa programot.

4.2. Indítsa el az XE251 fejlesztői környezetet! Olvassa be a programot (Fájl menü –Betölt-*.A51
). Aktualizálja a program fejlécét!

4.3. Fordítsa le a forrásnyelvű fájt és elemezze a listafájl felépítését.

4.4. A PC-hez csatlakoztatott mikrokontrolleres gyakorlóba töltse le a programot. A monitor prog-
ram parancsaival ellenőrizze a program unassemblált listáját, határozza meg a nyomógombok,
LED-ek bit címeit!

4.5. Indítsa el a programot és ellenőrizze a valós környezetben a működését!

4.6. A működést és tapasztalatait foglalja jegyzőkönyvbe!

5. Kérdések

5.1. Ismertesse az assembly programfejlesztés lépéseit!

5.2. Mi az assembly nyelv és mi a gépi kódú programozás?

5.3. Mi a szegmens?

5.4. Milyen szegmenseket használunk?

5.5. Mi az OBJ fájl?

5.6. Mi a szerepe a Linker-nek ?


