

## 2.MÉRÉS

### *PBC80C552 mikrokontroller-bázisú mikrogép programozása*

#### 1. A mérés célja:

Az XE2 fejlesztői környezet megismerése. A mikrogép hardver felépítésének, monitor programjának használata. Programfejlesztés, **programírás - fordítás - szerkesztés - futtatás** folyamat gyakorlása.

#### 2. A szükséges ismeretek

##### 2.1. A mikrogép hardver kialakítása

A mérési gyakorlatok a 80C552 típusú mikrokontrolleres gyakorló készüléken kell elvégezni. A készülék részletes leírása az **MK\_GY.DOC** fájlban van.

##### 2.2. Szoftver rendszer és fejlesztői környezet

A mikrokontrolleres **mikrogép - PC** osztott intelligenciájú rendszer. A programfejlesztéshez az **XE2** menüvezérelt fejlesztői környezetet használjuk amely a **C:/XE251** alkönyvtárban van. Indítása az **XE251.BAT** parancsfájllal történik. A környezetben mind assembly (\*.a51), mind pedig C (\*.c) nyelven végezhető a programfejlesztés.

Az indítás után megjelenő képen, az alábbi hat menüpont között lehet választani.

**Fájl   Editor   Assembler/Compiler   Tesztelés   Súgó   Kilépés/Quit**

A menüpontok kiválasztása mind az egérrel, mind pedig a cursor vezérlő nyilakkal történhet. (A fejlesztői környezet használatának részletes leírása a **PROG\_FEJL.DOC** állományban található.) Itt röviden leírtuk, hogy az egyes menükben milyen feladatok végezhetők.

A **File** menün belül a

- **Betölt** almenüben választható ki a fejlesztendő - már létező - forrásnyelvi fájl,
  - **\*.A51**      assembly nyelvű
  - **\*.C**        C nyelvűformája, illetve a
- **\*.PRJ**      project fájl.
- **\*.\***        tetszőleges kiterjesztésű állomány.
- **Megtekint** almenüben tekinthetők meg a fordítás során létrehozott
  - **Hiba-fájl**
  - **Lista-fájl**
  - **Map-fájl**
  - **Kevert C lista**
- **Takarít** almenüben törölhetők a fejlesztéskor létrejött fájlok.
- **Kilépés/Quit** almenün keresztül léphetünk ki a fejlesztői környezetből.

Az **Edit** menüen belül választható ki, majd írható, módosítható

- **Forrás** a kiválasztott forrásnyelvi fájl.
- **Project** a project fájl.

Az **Assembler/Compiler** menüben választható ki

- **Fordítás** a forrás-fájlok *assemblálása*, ill. *compillálása*. (Ekkor készülnek el a *\*.OBJ* - tárgykódú, és a *\*.LST* – lista fájlok.
- **Linker** *szerkesztés*, amely során az abszolút-object (futtatható), ill. a *\*.MAP* memória-térkép fájlok generálódnak.
- **Hex generálás** itt történik a futatható fájl letölthető formájává *\*.HEX* alakítása.

A **Linker** almenü kiválasztásakor egy további ablakban lehet kiválasztani az

- **Egyedi** alponban az egy modulus *assembly*-nyelvű,
- **Egyedi+Startup** alponban az egy modulus *C*-nyelvű,
- **Project** alponban a *\*.PRJ* parancsfájllal több modulból álló programok szerkesztését.

A **Tesztelés** menüben végezhető el az előzőek során hibátlanak bizonyult program ellenőrzése a

- **Szimulátor**  
almenüben szimulálása a *Simula5x* program segítségével. (Használata az *1. mérési útmutatóban* található).
- **Terminál**  
almenüben a **valós-idejű** ellenőrzés végezhető. A menüpont elindításakor jön létre - soros vonalon keresztül - a kommunikációs kapcsolat a **PC** és a **mikrogép** között. A kapcsolatot a PC-n futó *TVTERM.EXE*, és a mikrogépben futó **monitor** és **tesztelő** programok hozzák létre. (Használatát a *Mon\_haszn.doc* segédanyag írja le részletesen ).
- **Disassembler** megnyitásakor újabb ablakból választható ki
  - **Nyers forrás** a futtatható kódsorozat illetve,
  - **C-hivatkozásokkal** C nyelvű forrásnál a kevert visszafordítás ( disassemblálás, vagy unassemblálás).

A **Sugó** menü alatt különböző segédanyagok, pl. utasítás-lista stb. olvashatók.

### 2.3. A mikrogép monitor és tesztelő programja

A mikrogép **EPROM** - jában van a **monitor** és **tesztelő** program. A monitor segíti a fejlesztett program futtatását, ellenőrzését. Biztosítja a soros kommunikációt a terminálként használt PC-vel. Parancsai - a terminál funkcióban - a képernyőre kiíráthatók a **H** gomb lenyomásával.

## 3. Házi feladatok

- 3.1. Elemezze az alábbi *ELSO\_PR.A51* nevű programot, amely a mikrokontrolleres gyakorlón kiépített *nyomógombok*, és *LED* kijelzők használatára mutat példát! Készítse el a program *flyamatábráját*!

```

;*****
;*          BMF KKVFK  Automatika Intézet          *
;*          Elektronika Szakcsoport                *
;*****
;* Főprogram:      Első program                    *
;* Változat:       V1.0                            *
;* Dátum:          2003.07.20                      *
;* Készítette:     Zalotay Péter                   *
;*****
;* A modul leírása:                                *
;*      Az MK_GY mikrokontrolleres gyakorlón lévő *
;*      LED-ek és nyomgombok használata futó-fény *
;*      vezérlésére                                *
;*****

```

**NAME**        **Elso\_prog**                                ;a program modul neve

**;xxxxxxx DEKLARÁCIÓK xxxxxxxx**

**;\*\*\*\* Konstansok deklarációja \*\*\*\***

**IDO1**        **EQU 20H**                                ;a külső ciklus ismétlési száma

**IDO2**        **EQU 0A0H**                                ;a belső ciklus ismétlési száma

**;\*\*\*\* SZEGMENSEK DEKLARÁCIÓJA \*\*\*\***

**;++++ Relokálható szegmensek +++++**

**VALT**        **SEGMENT DATA**                        ;adatszegmens bájtos változóknak

**VALTB**       **SEGMENT DATA BITADDRESSABLE**  
    ;adatszegmens a bitcímezhető  
    ;területen lévő változóknak

**?STACK**      **SEGMENT IDATA**                        ;adatszegmens a ZSÁK-mem.részére

**PROG**        **SEGMENT CODE**                        ;program szegmens

**;++++ Abszolút címzésű szegmensek +++++**

**XSEG AT 0C000H**                                ;a memóriába ágyazott  
    ;perifériák

**NGS:**        **DS 1**                                ;külső memória-területen  
    ;nyomógomb-sor címe 0C000H

**LEDS:**       **DS 1**                                ;LED-sor címe 0C001H

**;++++ Bájtos változók szegmense +++++**

**RSEG VALT**

**DS 2**    ; helyfoglalás 2 bájt részére (csak példaként)

**KIM:**        **DS**    **1**                                ; helyfoglalás a kimeneti memória részére  
**M\_VALT:**   **DS**    **1**                                ; helyfoglalás a munka változó részére

;++++ **Bit-címezhető változók szegmense** +++++

**RSEG**        **VALTB**

**DS**        **2**                                ; helyfoglalás 2 bájt részére (csak példaként)  
**BEM: DS**    **1**                                ; helyfoglalás a bemeneti memória részére  
**ATM: DS**    **1**                                ; helyfoglalás az átmeneti tároló részére

; a bájtok egyes bitjeihez szimbólum rendelése

**BE\_NG**     **BIT**    **BEM.0**                        ;be-nyomógomb a **BEM** bájt **LSB** bit-je  
**KL\_NG**     **BIT**    **BEM.7**                        ;ki-nyomógomb a **BEM** bájt **MSB** bit-je  
**TART**      **BIT**    **ATM.0**                        ;tartó-bit az **ATM** bájt **LSB** bit-je

;++++ **A zsák (stack) szegmense** +++++

**RSEG ?STACK**

**DS**        **10H**

;++++ **A program szegmense** +++++

**RSEG PROG**

**LJMP**        **INIT**                                ;ugrás az inicializálásra

**DS**        **10H**                                ; megszakítás-ugrótáblázatnak helyfoglalás

**;xxxx INICIALIZÁLÁS xxxx**

**INIT: MOV**        **SP,#?STACK-1**                        ;az SP beállítása

;.... memória-nullázó for-ciklus ...

; ciklus-jellemzők megadása

**MOV**        **R0,#VALT**                                ;első bájt címének megadása

**MOV**        **R7,#20H**                                ;ciklus-változó a bájtok száma

; ciklus-törzs

**ISM: MOV**        **@R0,#0**                                ;a címzett bájtba 0 írása

**INC**        **R0**                                ;következő cím

; tesztelés

**DJNZ**        **R7,ISM**                                ;a ciklusváltozó csökkentése, és végellenőrzés

; kezdő-értékek beállítása

**MOV**        **M\_VALT,#80H**        ;a munka változó MSB-bitje 1

**MOV**        **DPTR,#LEDS**        ;a LED-sor sötétre vezérlése  
**MOV**        **A,#0FFH**  
**MOVX**       **@DPTR,A**

;xxxx FŐPROGRAM xxxx

**FOPR:**

;.... BEOLVAS ....

**MOV**        **DPTR,#NGS**        ;a nyomógombok aktuális értékének  
**MOVX**       **A,@DPTR**        ;beolvasása a BEM memóriába  
**CPL**        **A**  
**MOV**        **BEM,A**

;.... VEZÉRLÉS ....

**MOV**        **C,BE\_NG**        ;a TART-bit vezérlése a  
**ORL**        **C,TART**        ;  
**ANL**        **C,/KI\_NG**        ; $TART = (BE\_NG + TART) * KI\_NG$   
**MOV**        **TART,C**        ;logikai függvény szerint;(öntartás)

;.... DÖNTÉS ....

**JNB**        **TART,VEGE**        ;ha a TART nem 1, akkor áll a világító LED

;.... FORGATÁS ....

**MOV**        **A,M\_VALT**        ;TART=1 esetén a munka- változó tartalmának  
**RL**        **A**        ; forgatása balra  
**MOV**        **M\_VALT,A**

;.... KIIRATÁS ....

**MOV**        **DPTR,#LEDS**        ;az M\_VALT tartalmának kijelzése a LED-soron  
**MOV**        **A,M\_VALT**  
**CPL**        **A**  
**MOVX**       **@DPTR,A**

;.... KÉSLELTETÉS ....

**CALL**       **KESL**        ;a forgatás ismétlésének késleltetése

**VEGE:**

**JMP**        **FOPR**

;xxxx ALPROGRAMOK (szubrutinok) xxxx

;.... Szoftver-késleltető rutin ....  
; egymásba ágyazott ciklusokkal, az R7 és R6 ciklusváltozók

**KESL:**

**MOV R7,#IDO1** ;kezdőértékek beállítása

**KES1:**

**MOV R6,#IDO2**

**KES2:**

**DJNZ R6,KES2** ;belső ciklus

**DJNZ R7,KES1** ;külső ciklus

**RET** ;visszatérés a főprogramba

**END**

#### 4. Mérési feladatok:

**4.1.** Másolja át az *C:\XE251\MUNKA* könyvtáron belül létrehozott saját alkönyvtárába a *PELDAK* alkönyvtárban található *ELSO\_PR.A51* példa programot.

**4.2.** Indítsa el az XE251 fejlesztői környezetet! Olvassa be a programot (Fájl menü –Betölt-\*.A51). Aktualizálja a program fejlécét!

**4.3.** Fordítsa le a forrásnyelvű fájlt és elemezze a listafájl felépítését.

**4.4.** A PC-hez csatlakoztatott mikrokontrolleres gyakorlóba töltse le a programot. A monitor program parancsaival ellenőrizze a program unassemblált listáját, határozza meg a nyomógombok, LED-ek bit címeit!

**4.5.** Indítsa el a programot és ellenőrizze a valós környezetben a működését!

**4.6.** A működést és tapasztalatait foglalja jegyzőkönyvbe!

#### 5. Kérdések

**5.1.** Ismertesse az assembly programfejlesztés lépéseit!

**5.2.** Mi az assembly nyelv és mi a gépi kódú programozás?

**5.3.** Mi a szegmens?

**5.4.** Milyen szegmenseket használunk?

**5.5.** Mi az OBJ fájl?

**5.6.** Mi a szerepe a Linker-nek ?