
;******************************************************** 
;*  K K M F  Automatika Intézet   * 
;*    Elektronika Szakcsoport    * 
;******************************************************** 
;*  Főprogram:  FUT_MEG     * 
;*  Változat: v1.0     * 
;*  Dátum: 2001.12.12.        * 
;*  Készítette: ZP     * 
;******************************************************** 
;*  A modul leírása:     * 
;*   Futó-fény megszakításos vezérlése  * 
;* fénypont váltás 0,5 sec-ként   * 
;*       * 
;******************************************************** 
 
 ;xxxxxxx forditási parancsok xxxxxxx 
 
$XREF    ;Készűljön keresztreferencia lista 
$DEBUG    ;Készűljön memória-map 
 
NAME FO_MOD 
 
;################### DEKLARÁCIÓK ################################### 
 
 ;******* Hardver deklarációk ********** 
 
LEDS  XDATA 0C001H  ;LED sor címe 
NGS     XDATA 0C000H  ;Nyomogomb-sor címe 
 
UJMOD EQU 00000010B ;uj - 2-es - űzemmód  állandójának deklarálása 
    ;C/T=0 ( fo/12 számlálása, GATE=0 csak szoft- 
    ;ver indítás, M1 M0 = 1 0  ( 2 - es űzemmód ) 
 
KSZ EQU 164  ;100 mikrosec -os megszakításhoz tartozó 
    ;ujratöltési (reload) szám. 
 
IDO EQU 5000  ;a 0,5 sec-hoz tartozó megszakítások száma 
   
 ;******* Szegmensek deklarálása ****** 
 
PROG SEGMENT CODE   ;program 
VALTB   SEGMENT DATA BITADDRESSABLE   ;bitcímezhető memória 
STACK SEGMENT DATA    ;stack memória 
 
       ;******* ADAT szegmens bitcímezhető változóknak******* 
 
 RSEG VALTB 
 
KESL: DS 1   ;két bájt foglalása a késleltetés 
 DS 1   ;számlálásához 
BEM: DS 1   ;A nyomógombok beolvasott állapotát tároló mem. 
START BIT BEM.0  ;a futó fényt indító bit (ny.gomb_0) 



STOP BIT BEM.7   ;a futó fényt leállító bit (ny.gomb_7) 
 
KIM:  DS 1    ;A LED sorra kiviendő adat tárolója 
F_VALT: DS 1    ;A futó fény változójának tárolója 
JELZ:  DS 1    ;jelző bájt 
IB  BIT JELZ.0   ;az időzítés végét jelző bit 
FUT  BIT JELZ.7   ;fut st engedélyező bit 
 
 ;xxxxxx STACK kijelölés xxxxxxx 
 
 RSEG    STACK              ;STACK kezdet 
 DS      10H               ;16 bájtos STACK 
 
;################### PROGRAMOK ################################### 
 
 ;xxxxxx Program szegmens xxxxx 
 
 RSEG PROG 
 
KEZD: LJMP INIT   ;ugr s az inicializ l sra 
 
 ;****** Megszakít s ugrót bla ****** 
 
 DS      8   ;helyfoglal s a megszakit soknak 
 LJMP MEG_0   ;ugr s a T0 megszakít s rutinj ra 
 
 ;******* Inicializ l s ******* 
 
INIT: ANL TMOD,#0F0H  ;a T0 sz ml ló 2-es űzemmódj nak 
 ORL TMOD,#UJMOD  ;be llít sa 
 
 MOV TH0,#KSZ  ;megszakít si idő be llít sa 
 
 MOV KESL,#LOW IDO  ;a haszn lt memóri k    
 MOV KESL+1,#HIGH IDO ;kezdő értékeinek be llít sa 
 MOV F_VALT,#1 
 MOV BEM,#0 
 MOV KIM,#0 
 
 SETB TR0   ;T0 időzítő indít sa 
 
 SETB ET0   ;T0 megszakít s-kérés engedélyezése 
 SETB EA   ;Megszakít s elfogad s engedélyezése 
 
 ;******** Fő program ********* 
 
FOPROG:  
 MOV DPTR,#NGS  ;A nyomógombok aktu lis értkének 
 MOVX  A,@DPTR   ;beolvas sa és t rol sa olyan mó- 
 CPL A   ;don, hogy a lenyomott gombot je- 
 MOV BEM,A   ;lezze 1. 
 



 MOV C,START   ;Indít s-le llít s vezérlése 
 ORL C,FUT   ;öntartó be llít s 
 ANL C,/STOP 
 MOV FUT,C  
 
 JNB FUT,FOPROG  ;Ha nincs engedélyezve a fut s, akkor 
     ;ismételt beolvas s 
 
 MOV A,F_VALT  ;A v ltozó forgat sa 
 
 JBC IB,FORGAT  ;Ha letelt a 0,5 sec akkor forgat s, 
 JMP KIJ   ;egyébként csak kijelzés 
 
FORGAT: 
 RL A   ;Forgat s, t rol s 
 MOV F_VALT,A 
 
KIJ: CPL A 
 MOV DPTR,#LEDS  ;Kijelz s a LED sorra 
 MOVX @DPTR,A 
 
 JMP FOPROG 
 
 ;******** a megszakít s rutin ******** 
 
MEG_0: PUSH ACC   ;Mentések 
 PUSH PSW 
 DJNZ KESL,V1   ;a kisebb helyérték csökkentése 1-el, 
     ;amig nem 0 addig kilépés a rutinból, 
 DJNZ KESL+1,V2  ;egyébként a magasabb helyérték csök-  
     ;kentése, ha nem 0 akkor az alacso- 
     ;nyabb érték ujra kezdése, egyébként 
 SETB IB   ;az 0,5 sec-t jelző bit be llít sa,     
 MOV  KESL+1,#HIGH IDO ;és a késleltető b jtok kezdő értékének 
V2: MOV KESL,#LOW IDO  ;ujrír sa 
 
V1: POP PSW   ;Mentett értékek visszahív sa 
 POP ACC 
 RETI    ;Visszatérés 
 
 
 END 


